#

pkgname=file-roller
pkgver=2.32.2
pkgrel=1
pkgdesc="Create and modify archives"
arch=('ppc')
url="http://www.gnome.org"
license=('GPL')
depends=('glib2' 'gconf' 'gtk2' 'libsm' 'desktop-file-utils' \
				'hicolor-icon-theme')
makedepends=('intltool' 'pkg-config' 'gettext')
optdepends=('unrar: the RAR uncompression program'
						'unzip: unpacks .zip archives'
						'p7zip: 7zip compression utility'
						'xz: managing LZMA and XZ compressed files')
install=${pkgname}.install
source=(ftp://ftp.gnome.org/pub/GNOME/sources/${pkgname}/2.32/${pkgname}-${pkgver}.tar.bz2)
md5sums=('2ec596e237c2f3af1d7d3f5b9c12884f')

build() {
	cd "${srcdir}/${pkgname}-${pkgver}"
	
	sh configure --prefix=/usr \
		--sysconfdir=/etc \
		--libexecdir=/usr/lib/file-roller \
		--localstatedir=/var \
		--disable-static \
		--disable-nautilus-actions \
		--disable-packagekit \
		--enable-nls \
		--disable-scrollkeeper \
		--disable-schemas-install \
		--with-gtk=2.0 \
		--without-help-dir \
		--without-omf-dir

	make
}

package() {
	cd "${srcdir}/${pkgname}-${pkgver}"

	make DESTDIR="${pkgdir}" install
}
